RSS Tech. Memo. 052096 **Issued: May 20, 1996**

Interim Report for JPL Contract 960132

Final Delivery:

Climatology of 14-GHz Atmospheric Attenuation

Principal Investigator: Frank J. Wentz

Prepared for:
NSCAT Project
Jet Propulsion Laboratory
National Aeronautics and Space Administration
Pasadena, CA 91109

Prepared by:

Remote Sensing Systems

1101 College Ave., Suite 220, Santa Rosa, CA 95404 (707) 545-2904

1. Introduction

This document describes the final delivery of the 14-GHz atmospheric attenuation climatology for NSCAT. The initial production of this climatology is described in the following document, which henceforth is called *Wentz* [1995]:

> Wentz, F. J., *Climatology of 14-GHz Atmospheric Attenuation*, Remote Sensing Systems Tech. Memo. 093095, September, 30, 1995.

The initial climatology was based on 4 years of SSM/I observations, whereas the final climatology is based on 8 years of SSM/I observations: 1988 through 1995. There are a few other differences, which we will now describe. Note that throughout this report when referring to the attenuation, we will mean the <u>2-way, nadir attenuation in units of decibels</u>.

The 14-GHz attenuation consist of three parts: water vapor, oxygen, and cloud water. We do not attempt to include rain because it is highly variable in both space and time. In the absence of rain, water vapor is the major contributor to the atmospheric attenuation. The global water vapor field has persistent large scale features that can be characterized in terms of a climatology. Values for the columnar water vapor (in units of millimeters) are derived from the SSM/I observations as described in *Wentz* [1995] and *Wentz* [1996]. The SSM/I water vapor observations are averaged onto monthly, 1° latitude by 1° longitude maps to produce a climatology.

The years 1988 through 1995 are used to produce the climatology. The first step is to produce 96 monthly maps for this 8-year period. For months in which two SSM/I's are in operation, the water vapor retrievals from the two satellites are averaged together. The next step is to average over the 8 years for a given month. In doing this average, each year is given equal weight. Finally, equation (5) in *Wentz* [1995] is used to convert the columnar water vapor to a water vapor attenuation.

To obtain the total atmospheric attenuation, we add the oxygen attenuation and cloud attenuation to the water vapor attenuation. The oxygen attenuation varies little over the globe. For very cold (warm) air, its value is 0.10 dB (0.08 dB). Equation (4) in *Wentz* [1995] is used to computed the oxygen attenuation, in which the water vapor content is used as a proxy for the air temperature T_D.

In the absence of rain, the cloud attenuation ranges from 0 dB to a typical value of 0.08 dB for heavy, but non-raining, clouds having 0.3 mm of liquid water. Clouds, like rain, are highly variable on both space and time, and we do not attempt to characterize this variability in terms of a climatology. Rather, we simply compute a single mean global value for the cloud attenuation. The SSM/I observations give a mean global cloud water value of 0.1 mm, which corresponds to an attenuation of 0.03 dB. This constant 0.03 dB is added to the vapor and oxygen attenuations to obtain the total atmospheric attenuation. Fortunately, although the cloud attenuation is highly variable, it is quite small when there is no rain.

2. Extending the Attenuation Climatology to Land and Sea Ice Areas

The SSM/I water vapor retrieval can only be done over the oceans. We were concerned about abruptly ending the attenuation climatology near coastlines and ice edges. Accordingly, we decided to extend the attenuation climatology to land and ice areas. This was done by using the NASA Water Vapor Project Data Set (NVAP). This data set provides monthly water vapor maps that cover the entire globe. The over-land and over-ice water vapors are obtained by merging TOVS vapor retrievals with radiosonde reports. We averaged the NVAP data in the same way as described above for the SSM/I vapors and produced a monthly climatology. Note that the NVAP data set only covers the 5 years from 1988 through 1992. We then merged our over-ocean vapor contents with the NVAP over-land/ice values to produced global water vapor fields having no holes.

The following merging procedure is used, in which an open-ocean cell is defined as a 1° latitude/longitude cell containing valid SSM/I vapor retrievals. (Open ocean cells are at least 50 km from coastlines and ice edges.)

1. The SSM/I vapor is used for all open ocean cells.

2. If a given 1° latitude/longitude cell is more than 555 km (i.e., 5°) from an open ocean cell, then the NVAP vapor is used.

3. Otherwise, the following weighting scheme is used.

$$V = \frac{\omega_0 V_{NVAP} + \sum_{I=1}^{N} \omega_I V_{SSMI,I}}{\sum_{I=0}^{N} \omega_I} \tag{1}$$

where ω_I are the blending weights, V_{NVAP} is the NVAP vapor value for the specified cell, and $V_{SSMI,I}$ are the SSM/I vapor values for neighboring ocean cells. The summation is over all open ocean cells that are within 555 km of the specified cell. If there are no ocean cells within 555 km, then N = 0 and (1) reduces to $V = V_{NVAP}$. We want the blending to be highly local in the sense that coastline cells are assigned vapor values that are similar to the value for the nearest ocean cell. Hence we use a R^{-4} weighting, where R_I is the distance (km) from the specified cell to the I^{th} ocean cell. (Note that if the specified cell is an ocean cell, then $V = V_{SSMI}$ and (1) is not done.)

$$\omega_I = R_I^{-4} \tag{3}$$

For the V_{NVAP} weight ω_0, we use R_0 = 200 km. With this choice of weights, V is approximately an equal blend of V_{NVAP} and V_{SSMI} when the specified cell is 300 km inland from a straight coastline.

Global attenuation fields are computed from the global water vapor fields as described above. The result is a monthly 14-GHz attenuation climatology that completely covers the globe.

3. Implementation of the Attenuation Climatology

The monthly maps of 14-GHz atmospheric attenuation are stored in a binary file called CLIMATTN.DAT, which is on the enclosed floppy diskette. This file consists of 777,600 bytes, and it corresponds to the following CHARACTER*1 array: ABUF(360,180,12). The first dimension corresponds to longitude. The second dimension corresponds to latitude. The third dimension corresponds to month, starting with January. The following segment of Fortran code shows how to interpret the ABUF array.

```
      CHARACTER*1 ABUF(360,180,12)
      DO 100 IMON=1, 12
      DO  90 ILAT=1,180
      DO  80 ILON=1,360
      XLAT = –90.5 +ILAT                    ! latitude
      XLON = – 0.5 +ILON                    ! east longitude
      IVAL=ICHAR(ABUF(ILON,ILAT,IMON))
      A2WAYDB=0.002*IVAL  !TWO-WAY, NADIR ATTENUATION (dB)
   80 CONTINUE
   90 CONTINUE
  100 CONTINUE
```

In the above code, the variable A2WAYDB is the 14-GHz, two-way, nadir attenuation in decibels for the 1° cell having its center at latitude XLAT and east longitude XLON.

We are also providing an interface routine for implementing the attenuation climatology into the NSCAT processing. This routine is called FDATTN. The inputs for this routine are the following:

1. ITIME = seconds from beginning of year, range = 0 to 366*86400
2. XLAT = latitude (degrees), range = –90.0 to 90.0
3. XLON = east longitude (degrees), range = 0 to 360.0

Given these inputs, FDATTN does a tri-linear space-time interpolation and returns a value for the attenuation at the specified point. The source code for FDATTN is given on the next page and also on the enclosed floppy diskette. The source code is self-explanatory for easy implementation. The source code is contained in the file FDATTN.F. Note that the OPEN statement for CLIMATTN.DAT may need to be changed when the routine is implemented at JPL. Otherwise, the routine is standard Fortran 77 (or Fortran 90).

```
      SUBROUTINE FDATTN(ITIME,XLAT,XLON, ATTN,IERR)

C     This subroutine finds the 2-way, nadir attenuation (dB) at 14 GHz.
C     The attenuation is based on a climatology water vapor.
C     The attenuation includes atmospheric oxygen, water vapor, and nominal cloud
C     A mean global cloud cover of 0.1 mm is assumed
C     References:
C      Wentz, F. J., Climatology of 14-GHz Atmospheric Attenuation, Remote Sensing
C      Systems, September 30, 1995.
C      Wentz, F. J., Final Delivery: Climatology of 14-GHz Atmospheric Attenuation,
C      Remote Sensing Systems, May 20, 1996.
C      Contact:  Wentz@INDY.REMSS.COM

C     Inputs:
C      ITIME = seconds from begin of year, range = 0 to 366*86400
C      XLAT  = latitude (degrees), range = -90.0 to 90.0
C      XLON  = east longitude (degrees), range = 0 to 360.00
C     Output
C      ATTN  = 2-way, nadir attenuation (dB) at 14 GHz.
C      IERR  = error flag, 0 = no error, 1 = error occured

      INTEGER*4 ITIME,IERR
      REAL*4 XLAT,XLON,ATTN
      CHARACTER*1 ABUF(360,180,12)
      DATA ISTART/1/

C     BEGIN EXECUTION

      IF(ISTART.EQ.1) THEN
      ISTART=0
      OPEN(3,FILE='CLIMATTN.DAT',STATUS='OLD',
     1 ACCESS='SEQUENTIAL',FORM='BINARY')
      READ(3) ABUF
      CLOSE(3)
      ENDIF

C     CHECK INPUTS

      IF(ITIME.LT.0.OR.ITIME.GT.31622400) GO TO 900
      IF(XLON.LT.0..OR.XLON.GT.360.) GO TO 900
      IF(ABS(XLAT).GT.90.) GO TO 900

      IERR=0

C     DO TIME,LAT,LON INTERPOLATION
C     2629800 IS THE AVERAGE NUM SEC IN A MONTH, 86400*365.25/12

      BRIEF=(ITIME-1314900)/2629800.D0
      I1=INT(1+BRIEF)
      I2=I1+1
      A1=I1-BRIEF
      A2=1-A1
      IF(I1.EQ. 0) I1=12
      IF(I2.EQ.13) I2= 1
C
      BRIEF=XLAT+89.5
      J1=INT(1+BRIEF)
      J2=J1+1
      B1=J1-BRIEF
      B2=1-B1
      IF(J1.EQ.  0) J1=  1
      IF(J2.EQ.181) J2=180
C
      BRIEF=XLON-0.5
      K1=INT(1+BRIEF)
      K2=K1+1
      C1=K1-BRIEF
      C2=1-C1
      IF(K1.EQ.  0) K1=360
      IF(K2.EQ.361) K2=  1

      ATTN=0.002*
     1 (A1*B1*(C1*ICHAR(ABUF(K1,J1,I1))+C2*ICHAR(ABUF(K2,J1,I1)))+
     2  A1*B2*(C1*ICHAR(ABUF(K1,J2,I1))+C2*ICHAR(ABUF(K2,J2,I1)))+
     3  A2*B1*(C1*ICHAR(ABUF(K1,J1,I2))+C2*ICHAR(ABUF(K2,J1,I2)))+
     4  A2*B2*(C1*ICHAR(ABUF(K1,J2,I2))+C2*ICHAR(ABUF(K2,J2,I2))))
      IF(ATTN.LT.0. .OR. ATTN.GT.0.5) GO TO 900 !SHOULD NEVER HAPPEN
      RETURN

  900 CONTINUE
      IERR=1
      RETURN
      END
```

4. References

Wentz, F. J., *Climatology of 14-GHz Atmospheric Attenuation*, Remote Sensing Systems Tech. Memo. 093095, September, 30, 1995.

Wentz, F. J., A Well-Calibrated Ocean Algorithm for SSM/I, Accepted for publication *J. Geophysical Research*, 1996.